A Formal Proof of Cauchy's Residue Theorem

Wenda Li and Lawrence C. Paulson

Computer Laboratory, University of Cambridge
{wl302,lp15}@cam.ac.uk

Abstract. We present a formalization of Cauchy's residue theorem and two of its corollaries: the argument principle and Rouché's theorem. These results have applications to verify algorithms in computer algebra and demonstrate Isabelle/HOL's complex analysis library.

1 Introduction

Cauchy's residue theorem — along with its immediate consequences, the argument principle and Rouché's theorem — are important results for reasoning about isolated singularities and zeros of holomorphic functions in complex analysis. They are described in almost every textbook in complex analysis [3, 15, 16].

Our main motivation of this formalization is to certify the standard quantifier elimination procedure for real arithmetic: *cylindrical algebraic decomposition* [4]. Rouché's theorem can be used to verify a key step of this procedure: Collins' projection operation [8]. Moreover, Cauchy's residue theorem can be used to evaluate improper integrals like

$$\int_{-\infty}^{\infty} \frac{e^{itz}}{z^2+1} dz = \pi e^{-|t|}$$

Our main contribution[1] is two-fold:

- Our machine-assisted formalization of Cauchy's residue theorem and two of its corollaries is new, as far as we know.
- This paper also illustrates the second author's achievement of porting major analytic results, such as Cauchy's integral theorem and Cauchy's integral formula, from HOL Light [12].

The paper begins with some background on complex analysis (Sect. 2), followed by a proof of the residue theorem, then the argument principle and Rouché's theorem (3–5). Then there is a brief discussion of related work (Sect. 6) followed by conclusions (Sect. 7).

2 Background

We briefly introduce some basic complex analysis from Isabelle/HOL's Multivariate Analysis library. Most of the material in this section was first formalized in HOL Light by John Harrison [12] and later ported to Isabelle.

[1] Source is available from `https://bitbucket.org/liwenda1990/src_itp_2016/src`

2.1 Contour Integrals

Given a path γ, a map from the real interval $[0, 1]$ to $\mathbb{C}$, the contour integral of a complex-valued function f on γ can be defined as

$$\oint_\gamma f = \int_0^1 f(\gamma(t))\gamma'(t)dt.$$

Because integrals do not always exist, this notion is formalised as a relation:

```
definition has_contour_integral ::
    "(complex ⇒ complex) ⇒ complex ⇒ (real ⇒ complex) ⇒ bool"
        (infixr "has'_contour'_integral" 50)
  where "(f has_contour_integral i) g ≡
          ((λx. f(g x) * vector_derivative g (at x within {0..1}))
           has_integral i) {0..1}"
```

We can introduce an operator for the integral to use in situations when we know that the integral exists. This is analogous to the treatment of ordinary integrals, derivatives, etc., in HOL Light [12] as well as Isabelle/HOL.

2.2 Valid Path

In order to guarantee the existence of the contour integral, we need to place some restrictions on paths. A *valid path* is a piecewise continuously differentiable function on `[0..1]`. In plain English, the function must have a derivative on all but finitely many points, and this derivative must also be continuous.

```
definition piecewise_C1_differentiable_on
           :: "(real ⇒ 'a :: real_normed_vector) ⇒real set ⇒ bool"
          (infixr "piecewise'_C1'_differentiable'_on" 50)
  where "f piecewise_C1_differentiable_on i ≡
          continuous_on i f ∧
          (∃s. finite s ∧ (f C1_differentiable_on (i - s)))"

definition valid_path :: "(real ⇒ 'a :: real_normed_vector) ⇒ bool"
  where "valid_path f ≡ f piecewise_C1_differentiable_on {0..1::real}"
```

2.3 Winding Number

The winding number of the path γ at the point z is defined (following textbook definitions) as

$$n(\gamma, z) = \frac{1}{2\pi i} \oint_\gamma \frac{dw}{w - z}$$

A lemma to illustrate this definition is as follows:

```
lemma winding_number_valid_path:
  fixes γ::"real ⇒ complex" and z::complex
  assumes "valid_path γ" and "z ∉ path_image γ"
  shows "winding_number γ z
        = 1/(2*pi*i) * contour_integral γ (λw. 1/(w - z))"
```

2.4 Holomorphic Functions and Cauchy's Integral Theorem

A function is *holomorphic* if it is complex differentiable in a neighborhood of every point in its domain. The Isabelle/HOL version follows that of HOL Light:

```
definition holomorphic_on :: (infixl "(holomorphic'_on)" 50)    where
  "f holomorphic_on s ≡ ∀x∈s. f complex_differentiable (at x within s)"
```

As a starting point to reason about holomorphic functions, it is fortunate that John Harrison has made the effort to prove Cauchy's integral theorem in a rather general form:

```
theorem Cauchy_theorem_global:
  fixes s::"complex set" and f::"complex ⇒ complex"
    and γ::"real ⇒ complex"
  assumes "open s" and "f holomorphic_on s"
    and "valid_path γ" and "pathfinish γ = pathstart γ"
    and "path_image γ ⊆ s"
    and "⋀w. w ∉ s ⟹ winding_number γ w = 0"
  shows "(f has_contour_integral 0) γ"
```

Note, a more common statement of Cauchy's integral theorem requires the open set s to be simply connected (connected and without holes). Here, the simple connectedness is encoded by a homologous assumption

```
"⋀w. w ∉ s ⟹ winding_number γ w = 0"
```

The reason behind this homologous assumption is that a non-simply-connected set s should contain a cycle γ and a point a within one of its holes, such that `winding_number γ a` is non-zero. Statements of such homologous version of Cauchy's integral theorem can be found in standard texts[1, 15].

2.5 Remarks on the Porting Efforts

We have been translating the HOL Light proofs manually in order to make them more general and more legible. In the HOL Light library, all theorems are proved for $\mathbb{R}^n$, where n is a positive integer encoded as a type [14]. The type of complex numbers is identified with $\mathbb{R}^2$, and sometimes the type of real numbers must be coded as $\mathbb{R}^1$. Even worse, the ordered pair (x,y) must be coded, using complicated translations, as $\mathbb{R}^{m+n}$. We are able to eliminate virtually all mention of $\mathbb{R}^n$ in favour of more abstract notions such as topological or metric spaces. Moreover, our library consists of legible structured proofs, where the formal development is evident from the proof script alone.

3 Cauchy's Residue Theorem

As a result of Cauchy's integral theorem, if f is a holomorphic function on a simply connected open set s which contains a closed path γ, then

$$\oint_\gamma f(w) = 0$$

However, if the set s does have a hole, then Cauchy's integral theorem will not apply. For example, consider $f(w) = \frac{1}{w}$ so that f has a pole at $w = 0$, and γ is the circular path $\gamma(t) = e^{2\pi i t}$:

$$\oint_\gamma \frac{dw}{w} = \int_0^1 \frac{1}{e^{2\pi i t}} \left(\frac{d}{dt} e^{2\pi i t} \right) dt = \int_0^1 2\pi i dt = 2\pi i \neq 0$$

Cauchy's residue theorem applies when a function is holomorphic on an open set except for a finite number of points (i.e. isolated singularities):

```
lemma Residue_theorem:
  fixes s pts::"complex set" and f::"complex ⇒ complex"
    and γ::"real ⇒ complex"
  assumes "open s" and "connected s" and "finite pts" and
          "f holomorphic_on s - pts" and
          "valid_path γ" and
          "pathfinish γ = pathstart γ" and
          "path_image γ ⊆ s - pts" and
          "∀z. (z ∉ s) ⟶ winding_number γ z  = 0"
  shows "contour_integral γ f
       = 2 * pi * i *(∑p∈pts. winding_number γ p * residue f p)"
```

where `residue f p` denotes the residue of `f` at `p`, which we will describe in details in the next subsection.

Note, definitions and lemmas described from this section onwards are our original proofs (i.e. not ported from HOL Light) except where clearly noted.

3.1 Residue

A complex function f is defined to have an *isolated singularity* at point z, if f is holomorphic on an open disc centered at z but not at z.

We now define `residue f z` to be the path integral of f (divided by a constant $2\pi i$) along a small circular path around z:

```
definition residue::"(complex ⇒ complex) ⇒ complex ⇒ complex" where
  "residue f z = (SOME int. ∃e>0. ∀ε>0. ε<e
    ⟶ (f has_contour_integral 2 * pi * i * int) (circlepath z ε))"
```

To actually utilize our definition, we need not only to show the existence of such integral but also its invariance when the radius of the circular path becomes sufficiently small.

```
lemma base_residue:
  fixes s::"complex set" and f::"complex ⇒ complex"
    and e::real and z::complex
  assumes "open s" and "z ∈ s" and "e > 0"
    and "f holomorphic_on (s - {z})" and "cball z e ⊆ s"
  shows "(f has_contour_integral 2 * pi * i * residue f z) (circlepath z e)"
```

Here `cball` denotes the familiar concept of a closed ball:



Fig. 1. Circlepath c_e and c_ϵ around an isolated singularity z

```
definition cball :: "'a::metric_space ⇒ real ⇒ 'a set"
  where "cball x e = {y. dist x y ≤ e}"
```

Proof. Given two small circular path c_ϵ and c_e around z with radius ϵ and e respectively, we want to show that

$$\oint_{c_\epsilon} f = \oint_{c_e} f$$

Let γ is a line path from the end of c_e to the start of $-c_\epsilon$. As illustrated in Figure 1, consider the path

$$\Gamma = c_e + \gamma + (-c_\epsilon) + (-\gamma)$$

where $+$ is path concatenation, and $-c_\epsilon$ and $-\gamma$ are reverse paths of c_ϵ and γ respectively. As Γ is a valid closed path and f is holomorphic on the interior of Γ, we have

$$\oint_{\Gamma} f = \oint_{c_e} f + \oint_{\gamma} f + (-\oint_{c_\epsilon} f) + (-\oint_{\gamma} f) = \oint_{c_e} f - \oint_{c_\epsilon} f = 0$$

hence

$$\oint_{c_\epsilon} f = \oint_{c_e} f$$

and the proof is completed.

3.2 Generalization to a Finite Number of Singularities

The lemma `base_residue` can be viewed as a special case of the lemma `Residue_theorem` where there is only one singularity point and γ is a circular path. In this section, we will describe our proofs of generalizing the lemma `base_residue` to a plane with finite number of singularities.

First, we need the Stone-Weierstrass theorem, which approximates continuous functions on a compact set using polynomial functions.[2]

[2] Our formalization is based on a proof by Brosowski and Deutsch [7].

```
lemma Stone_Weierstrass_polynomial_function:
  fixes f :: "'a::euclidean_space ⇒ 'b::euclidean_space"
  assumes "compact s"
      and "continuous_on s f"
      and "0 < e"
  shows "∃g. polynomial_function g ∧ (∀x ∈ s. norm(f x - g x) < e)"
```

From the Stone-Weierstrass theorem, it follows that each open connected set is actually valid path connected (recall that our valid paths are piecewise continuously differentiable functions on the closed interval $[0, 1]$):

```
lemma connected_open_polynomial_connected:
  fixes s::"'a::euclidean_space set" and x y::'a
  assumes "open s" and "connected s" and "x ∈ s" and "y ∈ s"
  shows "∃g. polynomial_function g ∧ path_image g ⊆ s ∧
             pathstart g = x ∧ pathfinish g = y"
```

```
lemma valid_path_polynomial_function:
  fixes p::"real ⇒ 'a::euclidean_space"
  shows "polynomial_function p ⟹ valid_path p"
```

This yields a valid path γ on some connected punctured set such that a holomorphic function has an integral along γ:

```
lemma get_integrable_path:
  fixes s pts::"complex set" and a b::complex and f::"complex ⇒ complex"
  assumes "open s" and "connected (s - pts)" and "finite pts"
    and "f holomorphic_on (s - pts) "
    and "a ∈ s - pts" and "b ∈ s - pts"
  obtains γ where
    "valid_path γ" and "pathstart γ = a" and "pathfinish γ = b"
    and "path_image γ ⊆ s-pts" and "f contour_integrable_on γ"
```

Finally, we obtain a lemma that reduces the integral along γ to a sum of integrals over small circles around singularities:

```
lemma Cauchy_theorem_singularities:
  fixes s pts::"complex set" and f::"complex ⇒ complex"
    and γ::"real ⇒ complex" and h::"complex ⇒ real"
  assumes "open s" and "connected s" and "finite pts"
    and "f holomorphic_on (s - pts)" and "valid_path g"
    and "pathfinish g = pathstart g" and "path_image g ⊆ (s - pts)"
    and "∀z. (z ∉ s) ⟶ winding_number g z  = 0"
    and "∀p∈s. h p>0 ∧ (∀w∈cball p (h p). w∈s ∧ (w≠p ⟶ w ∉ pts))"
  shows "contour_integral g f = (∑p∈pts. winding_number g p
        * contour_integral (circlepath p (h p)) f)"
```

Proof. Since the number of singularities *pts* is finite, we do induction on them. Assuming the lemma holds when there are *pts* singularities, we aim to show the lemma for $\{q\} \cup pts$.

As illustrated in Figure 2, suppose c_q is a (small) circular path around q, by the lemma `get_integrable_path`, we can obtain a valid path γ' from the end of γ to the start of c_q such that f has an integral along γ'.



Fig. 2. Induction on the number of singularities

Consider the path

$$\Gamma = \gamma + \gamma' + \underbrace{c_q + ... + c_q}_{-n(\gamma,q)} + (-\gamma')$$

where + is path concatenation, $n(\gamma, q)$ is the winding number of the path γ around q and $-\gamma'$ is the reverse path of γ'. We can show that Γ is a valid cycle path and the induction hypothesis applies to Γ, that is

$$\oint_{\Gamma} f = \sum_{p \in pts} n(\gamma, p) \oint_{c_p} f$$

hence

$$\oint_{\gamma} f + \oint_{\gamma'} f - n(\gamma, q) \oint_{c_q} f - \oint_{\gamma'} f = \sum_{p \in pts} n(\gamma, p) \oint_{c_p} f$$

and finally

$$\oint_{\gamma} f = \sum_{p \in \{q\} \cup pts} n(\gamma, p) \oint_{c_p} f$$

which concludes the proof.

By combining the lemma `Cauchy_theorem_singularities` and `base_residue`, we can finish the proof of Cauchy's residue theorem (i.e. the lemma `Residue_theorem`).

3.3 Applications

Besides corollaries like the argument principle and Rouché's theorem, which we will describe later, Cauchy's residue theorem is useful when evaluating improper integrals.



Fig. 3. A semicircular path centered at 0 with radius $R > 1$

For example, evaluating an improper integral:

$$\int_{-\infty}^{\infty} \frac{dx}{x^2+1} = \pi$$

corresponds the following lemma in Isabelle/HOL:

lemma *improper_Ex:*
```
"Lim at_top (λR. integral {- R..R} (λx. 1 / (x² + 1))) = pi"
```

Proof. Let

$$f(z) = \frac{1}{z^2+1.}$$

Now $f(z)$ is holomorphic on $\mathbb{C}$ except for two poles when $z = i$ or $z = -i$. We can then construct a semicircular path $\gamma_R + C_R$, where γ_R is a line path from $-R$ to R and C_R is an arc from R to $-R$, as illustrated in Figure 3. From Cauchy's residue theorem, we obtain

$$\oint_{\gamma_R + C_R} f = 2\pi i \mathrm{Res}(f, i) = \pi$$

where $\mathrm{Res}(f, i)$ is the residue of f at i. Moreover, we have

$$\left| \oint_{C_R} f \right| \leq \frac{1}{R^2 - 1} \pi R$$

as $|f(z)|$ is bounded by $1/(R^2 - 1)$ when z is on C_R and R is large enough. Hence,

$$\oint_{C_R} f \to 0 \quad \text{when} \quad R \to \infty$$

and therefore

$$\int_{-\infty}^{\infty} \frac{dx}{x^2+1} = \oint_{\gamma_R} f = \oint_{\gamma_R + C_R} f = \pi \quad \text{when} \quad R \to \infty$$

which concludes the proof.

Evaluating such improper integrals was difficult for Avigad et al. [2] in their formalization of the Central Limit Theorem. We hope our development could facilitate such proofs in the future, though it may not be immediate as their proof is based on a different integration operator.

3.4 Remarks on the Formalization

It is surprising that we encountered difficulties when generalizing the lemma *base_residue* to the case of a finite number of poles. Several complex analysis textbooks [9, 16] omit proofs for this part (giving the impression that the proof is trivial). Our statement of the lemma *Cauchy_theorem_singularities* follows the statement of Theorem 2.4, Chapter IV of Lang [15], but we were reluctant to follow his proof of generalizing paths to chains for fear of complicating existing theories. In the end, we devised proofs for this lemma on our own with inspiration from Stein and Shakarchi's concept of a *keyhole* [16].

Another tricky part we have encountered is in the proof of the lemma *improper_Ex*. When showing

$$\oint_{\gamma_R + C_R} f = \mathrm{Res}(f, i) = \pi$$

it is necessary to show i ($-i$) is inside (outside) the semicircular path $\gamma_R + C_R$, that is,

$$n(i, \gamma_R + C_R) = 1 \wedge n(-i, \gamma_R + C_R) = 0$$

where n is the winding number operation. Such proof is straightforward for humans when looking at Figure 3. However, to formally prove it in Isabelle/HOL, we ended up manually constructing some ad-hoc counter examples and employed proof by contradiction several times. Partially due to this reason, our proof of the lemma *improper_Ex* is around 300 lines of code, which we believe can be improved in the future.

4 The Argument Principle

In complex analysis, the *argument principle* is a lemma to describe the difference between the number of zeros and poles of a meromorphic[3] function.

```
lemma argument_principle:
  fixes f h::"complex ⇒ complex" and poles s:: "complex set"
  defines "zeros≡{p. f p=0} - poles"
  assumes "open s" and "connected s" and
```

[3] holomorphic except for isolated poles

```
        "f holomorphic_on (s - poles)" and
        "h holomorphic_on s" and
        "valid_path γ" and
        "pathfinish γ = pathstart γ" and
        "path_image γ ⊆ s - (zeros ∪ poles)" and
        "∀z. (z ∉ s) ⟶ winding_number γ z = 0" and
        "finite (zeros ∪ poles)" and
        "∀p∈poles. is_pole f p"
  shows "contour_integral γ (λx. deriv f x * h x / f x) = 2 * pi * i *
        ((∑p∈zeros. winding_number γ p * h p * zorder f p)
         - (∑p∈poles. winding_number γ p * h p * porder f p))"
```

where

```
definition is_pole :: "('a::topological_space ⇒ 'b::real_normed_vector)
    ⇒ 'a ⇒ bool" where
  "is_pole f a =  (LIM x (at a). f x :> at_infinity)"
```

encodes the usual definition of poles (i.e. f approaches infinity as x approaches a). `zorder` and `porder` are the order of zeros and poles, which we will define in detail in the next subsection.

4.1 Zeros and Poles

A complex number z is referred as a *zero* of a holomorphic function f if $f(z) = 0$. And there is a local factorization property about $f(z)$:

```
lemma holomorphic_factor_zero_Ex1:
  fixes s::"complex set" and f::"complex ⇒ complex" and z::complex
  assumes "open s" and "connected s" and "z ∈ s" and "f(z) = 0"
      and "f holomorphic_on s" and "∃w∈s. f w ≠ 0"
  shows "∃!n. ∃g r. 0 < n ∧ 0 < r ∧ ball z r ⊆ s ∧
              g holomorphic_on ball z r
              ∧ (∀w∈ball z r. f w = (w-z)^n * g w ∧ g w≠0)"
```

Here a *ball*, as usual, is an open neighborhood centred on a given point:

```
definition ball :: "'a::metric_space ⇒ real ⇒ 'a set"
  where "ball x e = {y. dist x y < e}"
```

Proof. [4] As f is holomorphic, f has a power expansion locally around z:

$$f(w) = \sum_{k=0}^{\infty} a_k (w-z)^k$$

and since f does not vanish identically, there exists a smallest n such that $a_n \neq 0$. Therefore

$$f(w) = \sum_{k=n}^{\infty} a_k (w-z)^k = (w-z)^n \sum_{k=0}^{\infty} a_{k+n} (w-z)^k = (w-z)^n g(w)$$

[4] The existence proof of such `n, g and r` is ported from HOL Light, while we have shown the uniqueness of `n` on our own.

and the function $g(w)$ is holomorphic and non-vanishing near z due to $a_n \neq 0$.

Also, we can show that this n is unique, by assuming there exist m and another locally holomorphic function $h(w)$ such that

$$f(w) = (w - z)^n g(w) = (w - z)^m h(w)$$

and $h(w) \neq 0$. If $m > n$, then

$$g(w) = (w - z)^{m-n} h(w)$$

and this yields $g(w) \to 0$ when $w \to z$, which contradicts the fact that $g(w) \neq 0$. If $n > m$, then similarly $h(w) \to 0$ when $w \to z$, which contradicts $h(w) \neq 0$. Hence, $n = m$, and the proof is completed.

The unique `n` in the lemma *holomorphic_factor_zero_Ex1* is usually referred as the *order/multiplicity of the zero* of `f` at `z`:

```
definition zorder::"(complex ⇒ complex) ⇒ complex ⇒ nat" where
  "zorder f z = (THE n. n>0 ∧ (∃g r. r>0 ∧ g holomorphic_on cball z r
                  ∧ (∀w∈cball z r. f w =  g w * (w-z)^n ∧ g w ≠0)))"
```

We can also refer the complex function `g` in the lemma *holomorphic_factor_zero_Ex1* using Hilbert's epsilon operator in Isabelle/HOL:

```
definition zer_poly::"[complex ⇒ complex, complex]⇒ complex ⇒ complex"
    where
  "zer_poly f z = (SOME g. ∃r . r>0 ∧ g holomorphic_on cball z r
          ∧ (∀w∈cball z r. f w =  g w * (w-z)^(zorder f z) ∧ g w ≠0))"
```

Given a complex function f that has a pole at z and is also holomorphic near (but not at) z, we know the function

$$\lambda x. \quad \text{if } x = z \text{ then } 0 \text{ else } 1/f(x)$$

has a zero at z and is holomorphic near (and at) z. On the top of the definition of the order of zeros, we can define the *order/multiplicity of the pole* of f at z:

```
definition porder::"(complex ⇒ complex) ⇒ complex ⇒ nat" where
  "porder f z = (let f'=(λx. if x=z then 0 else inverse (f x))
                 in zorder f' z)"
```

```
definition pol_poly::"[complex ⇒ complex,complex]⇒complex ⇒ complex" where
  "pol_poly f z = (let f'=(λ x. if x=z then 0 else inverse (f x))
                   in inverse o zer_poly f' z)"
```

and a lemma to describe a similar relationship among `f`, `porder` and `pol_poly`:

```
lemma porder_exist:
  fixes f::"complex ⇒ complex" and s::"complex set"
    and z::complex
  defines "n≡porder f z" and "h≡pol_poly f z"
  assumes "open s" and "z ∈ s"
    and "f holomorphic_on (s - {z})"
    and "is_pole f z"
  shows "∃r. n>0 ∧ r>0 ∧ cball z r ⊆ s ∧ h holomorphic_on cball z r
    ∧ (∀w∈cball z r. (w≠z ⟶ f w  = h w / (w-z)^n) ∧ h w ≠0)"
```

Proof. With the lemma *holomorphic_factor_zero_Ex1*, we derive that there exist n and g such that

$$\text{if } w = z \text{ then } 0 \text{ else } 1/f(w) \;=\; (w-z)^n g(w)$$

and $g(w) \neq 0$ for w near z. Hence

$$f(w) = \frac{\frac{1}{g(w)}}{(w-z)^n} = \frac{h(w)}{(w-z)^n}$$

when $w \neq z$. Also, $h(w) \neq 0$ due to $g(w) \neq 0$. This concludes the proof.

Moreover, *porder* and *pol_poly* can be used to construct an alternative definition of residue when the singularity is a pole.

```
lemma residue_porder:
  fixes f::"complex ⇒ complex" and s::"complex set"
    and z::complex
  defines "n≡porder f z" and "h≡pol_poly f z"
  assumes "open s" and "z ∈ s"
    and "f holomorphic_on (s - {z})"
    and "is_pole f z"
  shows "residue f z = ((deriv ^^ (n - 1)) h z / fact (n - 1))"
```

Proof. The idea behind the lemma *residue_porder* is to view $f(w)$ as $h(w)/(w-z)^n$, hence the conclusion becomes

$$\frac{1}{2\pi i}\oint_{c_\epsilon} \frac{h(w)}{(w-z)^n} dw = \frac{1}{(n-1)!}\frac{d^{n-1}}{dw^{n-1}} h(z)$$

which can be then solved by Cauchy's integral formula.

4.2 The Main Proof

The main idea behind the proof of the lemma *argument_principle* is to exploit the local factorization properties at zeros and poles, and then apply the Residue theorem.

Proof (the argument principle). Suppose f has a zero of order m when $w = z$. Then $f(w) = (w-z)^m g(w)$ and $g(w) \neq 0$. Hence,

$$\frac{f'(w)}{f(w)} = \frac{m}{w-z} + \frac{g'(w)}{g(w)}$$

which leads to

$$\oint_\gamma \frac{f'(w)h(w)}{f(w)} = \oint_\gamma \frac{mh(w)}{w-z} = mh(z) \tag{1}$$

since

$$\lambda w. \frac{g'(w)h(w)}{g(w)}$$

is holomorphic (g, g' and h are holomorphic and $g(w) \neq 0$).

Similarly, if f has a pole of order m when $w = z$, then $f(w) = g(w)/(w-z)^m$ and $g(w) \neq 0$. Hence,

$$\oint_\gamma \frac{f'(w)h(w)}{f(w)} = \oint_\gamma \frac{-mh(w)}{w-z} = -mh(z) \tag{2}$$

By combining equations (1), (2) and the lemma *Cauchy_theorem_singularities* [5], we can show

$$\oint_\gamma \frac{f'(w)h(w)}{f(w)} = 2\pi i \left(\sum_{p \in \text{zeros}} n(\gamma,p)h(p)\text{zo}(f,p) - \sum_{p \in \text{poles}} n(\gamma,p)h(p)\text{po}(f,p) \right)$$

where $\text{zo}(f,p)$ (or $\text{po}(f,p)$) is the order of zero (or pole) of f at p, and the proof is now complete.

4.3 Remarks

Our definitions and lemmas in Section 4.1 roughly follow Stein and Shakarchi [16], with one major exception. When f has a pole of order n at z, Stein and Shakarchi define residue as

$$\text{Res}(f,z) = \lim_{w \to z} \frac{1}{(n-1)!} \frac{d^{n-1}}{dw^{n-1}} \left[(w-z)^n f(w) \right]$$

while our lemma *residue_porder* states

$$\text{Res}(f,z) = \frac{1}{(n-1)!} \frac{d^{n-1}}{dw^{n-1}} h(z)$$

where $f(w) = \frac{h(w)}{(w-z)^n}$ and $h(w)$ is holomorphic and non-vanishing near z. Note, $h(w) = (w-z)^n f(w)$ only when $w \neq z$, since $f(w)$ is a pole (i.e. undefined) when $w = z$. Introducing the function h eliminates the technical difficulties of reasoning about limits formally.

5 Rouché's Theorem

Given two functions f and g holomorphic on an open set containing a path γ, if

$$|f(w)| > |g(w)|$$

for all $w \in \gamma$, then Rouché's Theorem states that f and $f+g$ have the same number of zeros counted with multiplicity and weighted with winding number:

[5] Either the lemma *Cauchy_theorem_singularities* or the lemma *Residue_theorem* suffices in this place.

```
lemma Rouche_theorem:
  fixes f g::"complex ⇒ complex" and s:: "complex set"
  defines "fg≡(λp. f p+ g p)"
  defines "zeros_fg≡{p. fg p =0}" and "zeros_f≡{p. f p=0}"
  assumes "open s" and "connected s" and
    "finite zeros_fg" and "finite zeros_f" and
    "f holomorphic_on s" and "g holomorphic_on s" and
    "valid_path γ" and "pathfinish γ = pathstart γ" and
    "path_image γ ⊆ s" and
    "∀z∈path_image γ. cmod(f z) > cmod(g z)" and
    "∀z. (z ∉ s) ⟶ winding_number γ z = 0"
  shows "(∑p∈zeros_fg. winding_number γ p * zorder fg p)
        = (∑p∈zeros_f. winding_number γ p * zorder f p)"
```

Proof. Let $\mathbb{Z}(f+g)$ and $\mathbb{Z}(f)$ be the number of zeros that $f+g$ and f has respectively (counted with multiplicity and weighted with winding number). By the argument principle, we have

$$\mathbb{Z}(f+g) = \frac{1}{2\pi i}\oint_\gamma \frac{(f+g)'}{f+g} = \frac{1}{2\pi i}\oint_\gamma \frac{f'}{f} + \frac{1}{2\pi i}\oint_\gamma \frac{(1+\frac{g}{f})'}{1+\frac{g}{f}}$$

and

$$\mathbb{Z}(f) = \frac{1}{2\pi i}\oint_\gamma \frac{f'}{f}$$

Hence, $\mathbb{Z}(f+g) = \mathbb{Z}(f)$ holds if we manage to show

$$\frac{1}{2\pi i}\oint_\gamma \frac{(1+\frac{g}{f})'}{1+\frac{g}{f}} = 0.$$

As illustrated in Figure 4, let

$$h(w) = 1 + \frac{g(w)}{f(w).}$$

Then the image of $h \circ \gamma$ is located within the disc of radius 1 centred at 1, since $|f(w)| > |g(w)|$ for all w on the image of γ. In this case, it can be observed that 0 lies outside $h \circ \gamma$, which leads to

$$\oint_{h\circ\gamma} \frac{dw}{w} = n(h\circ\gamma, 0) = 0$$

where $n(h\circ\gamma, 0)$ is the winding number of $h\circ\gamma$ at 0. Hence, we have

$$\oint_\gamma \frac{(1+\frac{g}{f})'}{1+\frac{g}{f}} = \int_0^1 \frac{h'(\gamma(t))}{h(\gamma(t))}\gamma'(t)dt = \int_0^1 \frac{(h\circ\gamma)'(t)}{(h\circ\gamma)(t)}dt = \oint_{h\circ\gamma}\frac{dw}{w} = 0$$

which concludes the proof.

Our proof of the lemma `Rouche_theorem` follows informal textbook proofs [3, 15], but our formulation is more general: we do not require γ to be a regular closed path (i.e. where $n(\gamma, w) = 0 \vee n(\gamma, w) = 1$ for every complex number w that does not lie on the image of γ).



Fig. 4. The path image of $\lambda t.1 + \frac{g(\gamma(t))}{f(\gamma(t))}$ when $|f(w)| > |g(w)|$ for all w on the image of γ

6 Related Work

HOL Light has a comprehensive library of complex analysis, on top of which the prime number theorem, the Kepler conjecture and other impressive results have been formalized [11, 12, 13, 14]. A substantial portion of this library has been ported to Isabelle/HOL. It should be not hard to port our results to HOL Light.

Brunel [6] has described some non-constructive complex analysis in Coq, including a formalization of winding numbers. Also, there are other Coq libraries (mainly about real analysis), such as Coquelicot [5] and C-Corn [10]. However, as far as we know, Cauchy's integral theorem (which is the starting point of Cauchy's residue theorem) is not available in Coq yet.

7 Conclusion

We have described our formalization of Cauchy's residue theorem as well as two of its corollaries: the argument principle and Rouché's theorem. The proofs are drawn from multiple sources, but we were still obliged to devise some original proofs to fill the gaps. We hope our work will facilitate further work in formalizing complex analysis.

Acknowledgements. We are grateful to John Harrison for his insightful suggestions about mathematical formalization, and also to the anonymous reviewers for their useful comments on the first version of this paper. The first author was funded by the China Scholarship Council, via the CSC Cambridge Scholarship programme.

References

[1] Ahlfors, L.V.: Complex Analysis: An Introduction to the Theory of Analytic Functions of One Complex Variable. McGraw-Hill, New York (1966)

[2] Avigad, J., Hölzl, J., Serafin, L.: A formally verified proof of the central limit theorem. CoRR abs/1405.7012 (2014)

[3] Bak, J., Newman, D.: Complex Analysis. Springer (2010)

[4] Basu, S., Pollack, R., Roy, M.F.: Algorithms in Real Algebraic Geometry. Springer (2006)

[5] Boldo, S., Lelay, C., Melquiond, G.: Coquelicot: A user-friendly library of real analysis for Coq. Mathematics in Computer Science 9(1), 41–62 (2014)

[6] Brunel, A.: Non-constructive complex analysis in Coq. In: 18th International Workshop on Types for Proofs and Programs, TYPES 2011, September 8-11, 2011, Bergen, Norway. pp. 1–15 (2011)

[7] Bruno Brosowski, F.D.: An elementary proof of the Stone-Weierstrass theorem. Proceedings of the American Mathematical Society 81(1), 89–92 (1981)

[8] Caviness, B., Johnson, J.: Quantifier Elimination and Cylindrical Algebraic Decomposition. Springer (2012)

[9] Conway, J.B.: Functions of One Complex Variable, vol. 11. Springer, New York, second edn. (1978)

[10] Cruz-Filipe, L., Geuvers, H., Wiedijk, F.: C-CoRN, the constructive Coq repository at Nijmegen. In: Mathematical Knowledge Management. pp. 88–103. Springer (2004)

[11] Hales, T.C., Adams, M., Bauer, G., Dang, D.T., Harrison, J., Hoang, T.L., Kaliszyk, C., Magron, V., McLaughlin, S., Nguyen, T.T., Nguyen, T.Q., Nipkow, T., Obua, S., Pleso, J., Rute, J., Solovyev, A., Ta, A.H.T., Tran, T.N., Trieu, D.T., Urban, J., Vu, K.K., Zumkeller, R.: A formal proof of the Kepler conjecture. arXiv 1501.02155 (2015)

[12] Harrison, J.: Formalizing basic complex analysis. In: Matuszewski, R., Zalewska, A. (eds.) From Insight to Proof: Festschrift in Honour of Andrzej Trybulec. vol. 10(23), pp. 151–165. University of Białystok (2007)

[13] Harrison, J.: Formalizing an analytic proof of the Prime Number Theorem (dedicated to Mike Gordon on the occasion of his 60th birthday). Journal of Automated Reasoning 43, 243–261 (2009)

[14] Harrison, J.: The HOL Light theory of Euclidean space. Journal of Automated Reasoning 50, 173–190 (2013)

[15] Lang, S.: Complex Analysis. Springer (1993)

[16] Stein, E.M., Shakarchi, R.: Complex Analysis, vol. 2. Princeton University Press (2010)